getc
,
copy chr to next cell over
[->+>+<<]>>[-<<+>>]<
divide by 10
>+++++++++<[>>>+<<[>+>[-]<<-]>[<+>-]>[<<++++++++++>>>+<-]<<-<-]
>>>>[<<<<+>>>>-]<<<<>[-]<
copy chr to next cell over
<[->>+>+<<<]>>>[-<<<+>>>]<
divide by 100
>++++++++++++++++++++++++++++++++++++++++++++++++++
++++++++++++++++++++++++++++++++++++++++++++++++++<
[>>>+<<[>+>[-]<<-]>[<+>-]>[<<++++++++++++++++++++++++++++++++++++++++++++++++++
++++++++++++++++++++++++++++++++++++++++++++++++++>>>+<-]<<-<-]>>>>[<<<<+>>>>-]
<<<<>[-]<
print chr/100 digit && restore value
>++++++++[-<++++++>]<.>++++++++[-<------>]<
multiply chr/100 by 10
[->+<]>[-<++++++++++>]<
chr/10 minus (chr/100)*10 to get remainder (10s digit)
[-<->]<
print 10s digit
>++++++++[-<++++++>]<.[-]<
copy chr to new cell
[>+>+<<-]>>[-<<+>>]<
divide by 10
>+++++++++<[>>>+<<[>+>[-]<<-]>[<+>-]>[<<++++++++++>>>+<-]<<-<-]
>>>>[<<<<+>>>>-]<<<<>[-]<
multiply chr/10 by 10
[->+<]>[-<++++++++++>]<
chr minus (chr/10)*10 to get remainder (1s digit)
[-<->]<
print 1s digit
>++++++++[-<++++++>]<.
newline
[-]++++++++++.
made by chris lutz
